Matrix multiplication — a refresher

Clear some variables so we can use them symbolically.

```
Clear[a, b, c, d, e, f, g, h, i, x, y, z]
```

Matrix and vector notation

Matrices are straightforward in *Mathematica*. Here is a 3×2 matrix (the numbers indicate "rows"×columns). In *Mathematica*'s regular nested list format, the numbers refer to the number of elements at each level of nesting from the 'outside in'

```
MatrixForm[{{a, b}, {c, d}, {e, f}}]
```

$$\begin{pmatrix} a & b \\ c & d \\ e & f \end{pmatrix}$$

Here is a 2×3 matrix.

```
MatrixForm[{{a, b, c}, {d, e, f}}]
```

$$\begin{pmatrix} a & b & c \\ d & e & f \end{pmatrix}$$

Here is a 3×3 square matrix.

```
MatrixForm[{{a, b, c}, {d, e, f}, {g, h, i}}]
```

$$\begin{pmatrix} a & b & c \\ d & e & f \\ g & h & i \end{pmatrix}$$

Vectors are almost as simple. A vector can be defined as a matrix with one dimension set to one. The first two examples both use nested lists to produce a 1×3 and a 3×1 matrix, which we would call a *row vector* and a *column vector* respectively.

```
MatrixForm[{{x, y, z}}]
```

$$\begin{pmatrix} x & y & z \end{pmatrix}$$

```
MatrixForm[{{x}, {y}, {z}}]
```

$$\begin{pmatrix} x \\ y \\ z \end{pmatrix}$$

Mathematica quirk alert! Look at this example, which uses a simple, non-nested list.

```
MatrixForm[{x, y, z}]
```

$$\begin{pmatrix} x \\ y \\ z \end{pmatrix}$$

This indicates that in some ways, *Mathematica* treats a plain list as a *column* vector (rather than a row vector, which is what it looks like in ordinary `{x, y, z}` notation). But, it's not exactly the same thing, as we will see.

■ Matrix and vector multiplication

Let's multiply matrices.

```
squareMatrix = {{a, b, c}, {d, e, f}, {g, h, i}};
MatrixForm[squareMatrix]
```

$$\begin{pmatrix} a & b & c \\ d & e & f \\ g & h & i \end{pmatrix}$$

One way is called 'elementwise' multiplication, and occurs if you use the regular "*" symbol.

```
MatrixForm[squareMatrix * squareMatrix]
```

$$\begin{pmatrix} a^2 & b^2 & c^2 \\ d^2 & e^2 & f^2 \\ g^2 & h^2 & i^2 \end{pmatrix}$$

As you can see, each element is lined up with the corresponding element from the other matrix, and the multiplication applied to each pair. Obviously, this only works if the matrices are exactly the same shape.

The much more useful *matrix* multiplication uses the 'dot' operator (just enter it as a period). Here is what the operation looks like in matrix notation…

```
MatrixForm[squareMatrix].MatrixForm[squareMatrix]
```

$$\begin{pmatrix} a & b & c \\ d & e & f \\ g & h & i \end{pmatrix} . \begin{pmatrix} a & b & c \\ d & e & f \\ g & h & i \end{pmatrix}$$

…and here is the result.

```
MatrixForm[squareMatrix.squareMatrix]
```

$$\begin{pmatrix} a^2 + b d + c g & a b + b e + c h & a c + b f + c i \\ a d + d e + f g & b d + e^2 + f h & c d + e f + f i \\ a g + d h + g i & b g + e h + h i & c g + f h + i^2 \end{pmatrix}$$

The result is also a 3×3 matrix. The top-left element is obtained by multiplying the *first* row of the first matrix by the *first* column of the second matrix, element by element, and then adding up the result. The top-middle element is obtained by multiplying the *first* row of the first matrix by the *second* column of the second matrix, element by element, and then adding up the result. In other words, the {row, column} index of each element in the result gives the row and column of the first and second input matrices respectively that are combined.

If you think about it, you will see that this will be possible only as long as the number of columns in the first matrix equals the number of rows in the second matrix. Here is a 2×3 matrix multiplied by a 3×2 matrix:

```
MatrixForm[{{a, b, c}, {d, e, f}}].MatrixForm[{{a, b}, {c, d}, {e, f}}]
```

$$\begin{pmatrix} a & b & c \\ d & e & f \end{pmatrix} . \begin{pmatrix} a & b \\ c & d \\ e & f \end{pmatrix}$$

```
MatrixForm[{{a, b, c}, {d, e, f}}.{{a, b}, {c, d}, {e, f}}]
```

$$\begin{pmatrix} a^2 + b c + c e & a b + b d + c f \\ a d + c e + e f & b d + d e + f^2 \end{pmatrix}$$

The result is a 2×2 matrix. One can express a matrix multiplication in terms of dimensions: $2\times3\cdot3\times2 = 2\times2$. The inner dimensions must match, and the outer dimensions give the dimensions of the result! If the inner dimensions fail to match, the result will be an error:

```
MatrixForm[{{a, b}, {c, d}, {e, f}}].MatrixForm[{{a, b, c}, {d, e, f}, {g, h, i}}]
```

$$\begin{pmatrix} a & b \\ c & d \\ e & f \end{pmatrix} . \begin{pmatrix} a & b & c \\ d & e & f \\ g & h & i \end{pmatrix}$$

```
MatrixForm[{{a, b}, {c, d}, {e, f}}.{{a, b, c}, {d, e, f}, {g, h, i}}]
```

Dot::dotsh : Tensors {{a, b}, {c, d}, {e, f}} and {{a, b, c}, {d, e, f}, {g, h, i}} have incompatible shapes. ≫

```
{{a, b}, {c, d}, {e, f}}.{{a, b, c}, {d, e, f}, {g, h, i}}
```

Vectors are no different. here are some examples, using the full row vector and column vector specification.

$2\times3\cdot3\times1 = 2\times1$

```
MatrixForm[{{a, b, c}, {d, e, f}}].MatrixForm[{{a}, {b}, {c}}]
```

$$\begin{pmatrix} a & b & c \\ d & e & f \end{pmatrix} . \begin{pmatrix} a \\ b \\ c \end{pmatrix}$$

```
MatrixForm[{{a, b, c}, {d, e, f}}.{{a}, {b}, {c}}]
```

$$\begin{pmatrix} a^2 + b^2 + c^2 \\ a\,d + b\,e + c\,f \end{pmatrix}$$

$1\times3\cdot3\times1 = 1\times1$

```
MatrixForm[{{a, b, c}}].MatrixForm[{{a}, {b}, {c}}]
```

$$\begin{pmatrix} a & b & c \end{pmatrix} . \begin{pmatrix} a \\ b \\ c \end{pmatrix}$$

```
MatrixForm[{{a, b, c}}.{{a}, {b}, {c}}]
```

$$\begin{pmatrix} a^2 + b^2 + c^2 \end{pmatrix}$$

$3\times1\cdot1\times3 = 3\times3$

```
MatrixForm[{{a}, {b}, {c}}].MatrixForm[{{a, b, c}}]
```

$$\begin{pmatrix} a \\ b \\ c \end{pmatrix} . \begin{pmatrix} a & b & c \end{pmatrix}$$

```
MatrixForm[{{a}, {b}, {c}}.{{a, b, c}}]
```

$$\begin{pmatrix} a^2 & a\,b & a\,c \\ a\,b & b^2 & b\,c \\ a\,c & b\,c & c^2 \end{pmatrix}$$